/PROG  PNS0189
/ATTR
OWNER		= MNEDITOR;
COMMENT		= "53379724TJ ";
PROG_SIZE	= 3337;
CREATE		= DATE 22-01-10  TIME 10:53:48;
MODIFIED	= DATE 25-02-25  TIME 22:00:22;
FILE_NAME	= PNS0062;
VERSION		= 0;
LINE_COUNT	= 118;
MEMORY_SIZE	= 3793;
PROTECT		= READ_WRITE;
TCD:  STACK_SIZE	= 0,
      TASK_PRIORITY	= 50,
      TIME_SLICE	= 0,
      BUSY_LAMP_OFF	= 0,
      ABORT_REQUEST	= 0,
      PAUSE_REQUEST	= 0;
DEFAULT_GROUP	= 1,*,*,*,*;
CONTROL_CODE	= 00000000 00000000;
/APPL
/MN
   1:  !******************************** ;
   2:  !     COSMA METAL FORMING  ;
   3:  !******************************** ;
   4:  !  LINE: CHINA CHANGSHA ;
   5:  !  ROBOT: R12 ;
   6:  !  PROGRAM: PNS0062 ;
   7:  !  PART:53379724  ;
   8:  !XEB0728************************* ;
   9:   ;
  10:  !INITIALIZE ;
  11:  MOVE TO HOME    ;
  12:  R[14:CG SENSETIVE]=75    ;
  13:  R[18:PICK TIME BEFORE]=1    ;
  14:  R[19:DROP TIME BEFORE]=.2    ;
  15:   ;
  16:  !------------------------------- ;
  17:  !BEGINNING OF PROGRAM ;
  18:  CALL PARTDROP    ;
  19:  COL GUARD ADJUST R[14] ;
  20:  UTOOL_NUM=1 ;
  21:  UFRAME_NUM=0 ;
  22:  PAYLOAD[3] ;
  23:  R[5:SPEED]=10    ;
  24:  OVERRIDE=R[5:SPEED] ;
  25:  !INITIATE TO PICK-POUNCE ;
  26:J P[1:MIDDLE] 50% CNT100 ACC100    ;
  27:J P[6] 50% CNT100 ACC100 BREAK    ;
  28:  PAUSE ;
  29:  R[1:CYCLE TOTAL]=0    ;
  30:  TIMER[1]=RESET ;
  31:  TIMER[1]=START ;
  32:   ;
  33:  LBL[10:TOP OF LOOF] ;
  34:  OVERRIDE=R[5:SPEED] ;
  35:   ;
  36:  TIMER[3]=STOP ;
  37:  TIMER[2]=STOP ;
  38:  R[1:CYCLE TOTAL]=R[1:CYCLE TOTAL]+1    ;
  39:  R[2:LAST CYCLE TIME]=TIMER[2]    ;
  40:  R[3:LAST SPM]=60/R[2:LAST CYCLE TIME]    ;
  41:  R[8:R MOVE TIME]=TIMER[3]    ;
  42:  TIMER[2]=RESET ;
  43:  TIMER[2]=START ;
  44:  TIMER[3]=RESET ;
  45:   ;
  46:  !PROCEED TO PICK ;
  47:  LBL[20] ;
  48:  IF DI[1:P1 PICK ENTRY OK]=ON AND DI[3:FROM RL1 CLEAR]=ON AND DI[8:P1 PICK START]=ON,JMP LBL[25] ;
  49:  IF DI[29:CYCLE STOP]=ON,JMP LBL[90] ;
  50:  JMP LBL[20] ;
  51:   ;
  52:  LBL[25] ;
  53:  R[5:SPEED]=100    ;
  54:  TIMER[3]=START ;
  55:  OVERRIDE=R[5:SPEED] ;
  56:J P[3:PICK APPROACH] 100% CNT70 ACC100    ;
  57:   ;
  58:J P[4:PICK] 100% CNT0 TB R[18]sec,CALL PARTPICK ACC100 BREAK    ;
  59:  PAYLOAD[4] ;
  60:  CALL CHK_PICK    ;
  61:  MONITOR Z_CHECH ;
  62:J P[5:PICK DEPART] 100% CNT70 ACC100 TA R[20]sec,DO[5:TO RL1 DROP START]=PULSE,1.0sec    ;
  63:  DO[8:P1 PARTS OUT]=PULSE,0.7sec ;
  64:   ;
  65:   ;
  66:J P[7:PICK CLEAR] 100% CNT100 ACC100    ;
  67:   ;
  68:  LBL[30:GOTO DROP POUNCE] ;
  69:J P[8:DROP POUNCE] 100% CNT100 ACC100 BREAK    ;
  70:   ;
  71:  TIMER[3]=STOP ;
  72:  LBL[40:AT DROP POUNCE] ;
  73:  IF DI[2:P2 DROP ENTRY]=ON AND DI[4:FROM R23 CLEAR]=ON AND DI[5:P2 DROP START]=OFF,JMP LBL[80] ;
  74:  IF DI[7:PRESS STAMPING COMPLETE]=ON AND DI[2:P2 DROP ENTRY]=ON,JMP LBL[50] ;
  75:  IF DI[18:MOVE TO CLEAR POS]=ON,JMP LBL[60] ;
  76:  JMP LBL[40] ;
  77:   ;
  78:  LBL[50:MOVE DROP SNEAK] ;
  79:  TIMER[3]=START ;
  80:  OVERRIDE=R[5:SPEED] ;
  81:J P[9:DROP SNEAK] 100% CNT100 ACC100 BREAK    ;
  82:  LBL[51:DROP OR C-P-S] ;
  83:   ;
  84:  IF DI[2:P2 DROP ENTRY]=ON AND DI[4:FROM R23 CLEAR]=ON AND DI[5:P2 DROP START]=ON,JMP LBL[80] ;
  85:  IF DI[18:MOVE TO CLEAR POS]=ON,JMP LBL[40] ;
  86:  JMP LBL[51] ;
  87:  JMP LBL[50] ;
  88:   ;
  89:  LBL[60:MOVE CLS-PRS-ACC] ;
  90:J P[1:MIDDLE] 50% CNT100 ACC100    ;
  91:  PAUSE ;
  92:  JMP LBL[40] ;
  93:   ;
  94:  LBL[80:DROP ENABLED] ;
  95:  TIMER[3]=START ;
  96:  !PROCEED TO DROP ;
  97:  OVERRIDE=R[5:SPEED] ;
  98:J P[10:DROP APPROACH] 100% CNT40 ACC100    ;
  99:  MONITOR END Z_CHECH ;
 100:J P[11:DROP] 100% CNT5 TB R[19]sec,CALL PARTDROP ACC100 BREAK    ;
 101:  DO[6:PART DROP COMPLETE]=PULSE,1.0sec ;
 102:  PAYLOAD[3] ;
 103:  CALL CHK_DROP    ;
 104:   ;
 105:J P[12:DROP DEPART] 100% CNT100 ACC130 TA   0.00sec,DO[7:PRESS STAMPING START]=PULSE,1.0sec    ;
 106:J P[13:DROP CLEAR] 100% CNT100 ACC130    ;
 107:  !GOTO PICK POUNCE ;
 108:J P[2:PICK POUNCE] 100% CNT100 ACC100 BREAK    ;
 109:  IF DI[29:CYCLE STOP]=ON,JMP LBL[90] ;
 110:  JMP LBL[10] ;
 111:   ;
 112:  LBL[90:END OF CYCLE] ;
 113:  TIMER[1]=STOP ;
 114:  R[4:TOTAL CYCLE TIME]=TIMER[1]    ;
 115:J P[1:MIDDLE] 50% CNT100    ;
 116:  MOVE TO HOME    ;
 117:   ;
 118:  LBL[100:END OF PROGRAM] ;
/POS
P[1:"MIDDLE"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2086.456  mm,	Y =     -.007  mm,	Z = -1111.822  mm,
	W =  -180.000 deg,	P =     -.000 deg,	R =  -180.000 deg,
	E1=   107.700 deg
};
P[2:"PICK POUNCE"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2072.359  mm,	Y =  1576.000  mm,	Z = -1136.206  mm,
	W =   179.321 deg,	P =     1.370 deg,	R =   150.805 deg,
	E1=   134.033 deg
};
P[3:"PICK APPROACH"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2635.869  mm,	Y =  3208.902  mm,	Z =  -937.956  mm,
	W =   179.015 deg,	P =    -1.744 deg,	R =    88.094 deg,
	E1=   195.893 deg
};
P[4:"PICK"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2635.868  mm,	Y =  3208.902  mm,	Z = -1107.795  mm,
	W =   179.015 deg,	P =    -1.744 deg,	R =    88.094 deg,
	E1=   195.893 deg
};
P[5:"PICK DEPART"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2635.869  mm,	Y =  3208.902  mm,	Z =  -880.356  mm,
	W =   179.015 deg,	P =    -1.744 deg,	R =    88.094 deg,
	E1=   195.893 deg
};
P[6]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2072.360  mm,	Y =  1575.890  mm,	Z =  -988.000  mm,
	W =   179.322 deg,	P =     1.369 deg,	R =   150.805 deg,
	E1=   134.032 deg
};
P[7:"PICK CLEAR"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2111.290  mm,	Y =  1815.421  mm,	Z =  -962.493  mm,
	W =   179.390 deg,	P =     -.076 deg,	R =   137.679 deg,
	E1=   146.540 deg
};
P[8:"DROP POUNCE"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  1932.596  mm,	Y =   511.598  mm,	Z =  -988.703  mm,
	W =   179.588 deg,	P =      .519 deg,	R =   168.496 deg,
	E1=   115.401 deg
};
P[9:"DROP SNEAK"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  1920.692  mm,	Y =   231.533  mm,	Z =  -993.185  mm,
	W =   179.693 deg,	P =      .605 deg,	R =   174.694 deg,
	E1=   109.139 deg
};
P[10:"DROP APPROACH"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  1686.931  mm,	Y =   231.533  mm,	Z = -1785.666  mm,
	W =   179.693 deg,	P =      .604 deg,	R =   174.694 deg,
	E1=   109.139 deg
};
P[11:"DROP"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  1686.931  mm,	Y =   231.533  mm,	Z = -1871.105  mm,
	W =   179.693 deg,	P =      .604 deg,	R =   174.694 deg,
	E1=   109.139 deg
};
P[12:"DROP DEPART"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  1877.972  mm,	Y =   231.533  mm,	Z = -1584.545  mm,
	W =   179.693 deg,	P =      .605 deg,	R =   174.694 deg,
	E1=   109.139 deg
};
P[13:"DROP CLEAR"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  1877.972  mm,	Y =   231.533  mm,	Z = -1584.545  mm,
	W =   179.693 deg,	P =      .605 deg,	R =   174.694 deg,
	E1=   109.139 deg
};
/END
